Environment Canada Imaging Cover Page

Report N.: 

SKP Box Number: 672672428

ATMOSPHERIC ENVIRONMENT SERVICE
DEPARTMENT OF THE ENVIRONMENT – CANADA

Technical Memoranda

A BASIC PROGRAM FOR THE MANIPULATION OF BIBLIOGRAPHIC INFORMATION BY A SMALL RESEARCH GROUP

by

M.J. CURRY

U.D.C.: 016.681.14
CN ISSN 0068-7804

TEC 798
14 DECEMBER 1973

ENVIRONMENT CANADA - ATMOSPHERIC ENVIRONMENT SERVICE
4905 Dufferin Street,
Downsview, Ontario.

A BASIC PROGRAM FOR THE MANIPULATION OF BIBLIOGRAPHIC INFORMATION BY A SMALL RESEARCH GROUP

by

M.J. Curry

ABSTRACT

A computer program is described, written in the BASIC language, which performs search-and-retrieve operations on a personalized bibliographic data file using a time-sharing computer or a mini-computer. It is shown that the program provides a convenient and economical way of cataloguing and accessing reference literature of interest to an individual or a small research group. The design philosophy of the program is briefly discussed, and examples of its operation are given.

UN PROGRAMME "BASIC" POUR MANIPULATION DE RENSEIGNEMENTS BIBLIOGRAPHIQUES PAR UN PETIT GROUPE DE CHERCHEURS

par

M.J. Curry

RÉSUMÉ

Un programme d'ordinateur est décrit dans le langage "BASIC". Il accomplit des opérations de recherche et relèvement sur un dossier personnel de données bibliographiques au moyen d'un ordinateur à utilisation collective ou d'un mini-ordinateur. L'on démontre que le programme dispense une façon pratique et économique de cataloguer et d'accéder à une littérature de reférence d'intérêt pour un individu ou un petit groupe de chercheurs. La philosophie de l'élaboration du programme est discutée brièvement, et des exemples relatifs à son opération sont donnés.

A BASIC PROGRAM FOR THE MANIPULATION OF BIBLIOGRAPHIC INFORMATION BY A SMALL RESEARCH GROUP.

by

Michael J. Curry

(Manuscript received December 13, 1973)

1. Introduction

One of the major problems facing an individual scientist or a small research team is that of maintaining adequate access to the open scientific literature. The problem is compounded not only by the sheer volume of published work but also by the increasingly multi-disciplinary nature of modern scientific research. For example, a research team working in the field of noise pollution may find it necessary to scan not only the traditional journals of physics, applied mathematics, and engineering but also some of the literature pertaining to such disciplines as biophysics, medicine, psychology, sociology, business and law.

Clearly, the task is a formidable one, although the scanning of current scientific literature has been greatly facilitated by the introduction of large-scale centralized information retrieval systems which are available for use by individuals and groups at reasonable cost. Such computerized scanning systems, of which the CAN/SDI* system is an example, permit the researcher to effortlessly and economically search a vast number of current journals in a wide variety of disciplines and to extract useful information on a number of independent or inter-related topics.

Whether a particular bibliography is compiled by such automated means or by the sheer expenditure of man-hours in a library, a second problem quickly emerges, viz: what to do with the accumulated bibliographic data to render them both tractable and accessible in a fairly flexible way. It is not unreasonable to suppose that a small research group (or even an individual) could very quickly conglomerate several thousand references having direct or indirect relevance to the research problem at hand. Of these items, perhaps 50 to 75 percent might be available in local libraries, while the research group itself might possess 10 percent in hard-copy or microcopy form. It quickly becomes a major task to simply provide the capability to locate a particular item at need.

* The CAN/SDI system is a computerized scientific information retrieval system operated by the National Science Library of the National Research Council of Canada.

Subject classification of such bibliographic data can be an even more awesome task. Each of the several thousand papers in the bibliography may deal with several topics or aspects of a topic. Many may have multiple authors. Clearly, a fully cross-referenced author and subject card file would require tens of thousands of cards. Such a system is not only unwieldy, it is also expensive and time-consuming to establish.

Most researchers have a convenient solution to these problems readily at hand in the form of either time-sharing computer systems accessible through remote, telephone-linked terminals or privately owned mini-computers. All that is required is to store the accumulated bibliographic information in a pre-determined way on magnetic tape or disc and to compose a suitable program which will (a) search the bibliography and retrieve selected information on demand, and (b) provide the information necessary to quickly and conveniently locate a copy of a particular paper.

This report describes one such program. Written in the BASIC language, it was designed to fulfill the requirements of the Atmospheric Acoustics section of the Atmospheric Environment Service but may conceivably be adapted to suit the needs of others. It is certainly not the only such program in existence and, since no attempt has been made to survey the extensive literature in the field of information retrieval, it cannot be acclaimed the best. It is, however, easy to use and fairly flexible and offers a convenient and inexpensive solution to the information handling problems discussed above.

This program has been designed for use with a Digital Equipment Corporation PDP-10 time-sharing computer system but could presumably be adapted for other computer systems which provide operating facilities for Extended BASIC. The program itself is presented in Appendix A and a sample data file in Appendix B. An additional program, called CHECK, has been written as an aid to the debugging of new data files. This program is described in Appendix D.

2. Program Concept and Design

(a) Choice of Language: The decision to program in BASIC was prompted by several considerations. The first and foremost was a desire to escape from the rigid format restrictions of FORTRAN. The second and perhaps no less important, is the ease with which alphanumeric ("string") data can be transmitted and manipulated in BASIC. Another is that, since BASIC was designed as a conversational language for use by people who are relatively inexperienced in computer techniques, it is a very "forgiving" piece of software, providing re-

cognizable and easily handled recoveries from user errors. The fact that BASIC is a very popular language for use with mini-computers is considered to be a coincidental bonus.

(b) The Nature of the Data: The data to be processed by the program consist of bibliographic information pertaining to scientific papers, books, theses and other communications. The relevant details concerning such items can be divided into seven groups or fields.

(i) Author - This field contains the name(s) of the author(s) of the work and may include such details as his institutional affiliation or mailing address.

(ii) Title - The title of the work may include, for example, a subtitle or information about the translation into English.

(iii) Reference - In the case of a journal article, the reference is composed of the name and volume of the journal, the paging of the paper, and the date of publication. In the case of books, theses and reports, this field contains information concerning the publishing agency and the date of publication.

(iv) Abstract - This field contains the author abstract of a paper or thesis, the information abstract as provided by an abstracting service or computerized retrieval system, or a summary of the contents of the referenced item.

(v) Subject Code Numbers - These are numbers which are assigned by the user(s) of the program to assist in the retrieval of bibliographic items which treat a particular topic or sub-topic.

(vi) Accession Number - This may be a file number by which the referenced item may be located in the files of the user scientist or research group, or it may be a code number corresponding to a message which indicates where the item may be found.

(vii) Remarks - This is a "catch-all" field which may contain comments on the usefulness of the referenced item in the context of the current research problems, information regarding previous or subsequent papers on the same subject, and so on.

It is assumed that the first three of these fields will be present in all cases. Null field provisions are made in the case of the remaining four.

(c) Organization of the Data: For purposes of definition, a number of restrictions have been imposed upon the data formats. It should be kept in mind that these limits are somewhat arbitrary and could be modified if necessary.

BASIC requires that string data which contain commas or certain other special characters be enclosed in quotation marks. Thus, in general, each line of text in the data file must be contained within quotation marks. This rule applies only to alphanumeric data and thus does not apply to the subject code numbers or accession number. It follows that, since quotation marks are used as string delimiters, they cannot appear within the text. Single quotes may be substituted for double quotes within the body of an alphanumeric field.

An exalamation mark (!) is used as a field delimiter in the present program. Operationally, this means that the final line in each of fields 2, 3, 4 and 7 must end in the symbol combination !".

Details of the particular code group scheme used by the author are given in Appendix C. In general, each code group consists of four numbers, each number representing a successively finer sub-division of subject matter. For example, the code group 2, 2, 5, 1 might represent the subject combination ACOUSTICS, NOISE, PROPAGATION, INSTRUMENTS, signifying that this particular paper deals, at least in part, with instrumentation for the study of the propagation of noise. Zeros within the code groups are wild. For example, a very general textbook on acoustics might be coded 2, 0, 0, 0 while a command to search for the combination 1, 0, 0, 2 would retrieve all papers which had 1 as the first number and 2 as the last number of a code group, regardless of the intervening digits. Clearly, a zero as the first number in a code group is not meaningful. Therefore, a zero in this position is used as a delimiter when the code groups are read into the program.

The data for each item in the bibliography, then, are arranged in the following way and in the order shown:

(i) Author - A single line of text, enclosed in quotation marks, containing the author's name and possibly other information It is suggested that a standard format be used for entering all authors' names - for example, surname first, followed by initials with appropriate punctuation, as SMITH, A.A.

(ii) Title - Up to five lines of text, each line enclosed in quotation marks, with the final line terminated with an exclamation mark.

(iii) Reference - Up to five lines of text, each line enclosed in quotation marks. The final line must end in the year of publication, followed by an exclamation mark. Thus a typical reference group would end in the combination 1973!".

(iv) Abstract - Up to twenty lines of text, each line enclosed in quotation marks, with the final line terminated in an exclamation mark. If no abstract is available, this field simply consists of the symbol combination "!".

(v) Subject Code Groups - Up to five groups are permitted, each group consisting of four numbers. Zeros are wild in the last three numbers in a group but are not recognized as legal first numbers. The final code group is followed by a single zero which acts as a terminator for this field. If no code groups have been assigned to a particular paper, this group consists of a single zero.

(vi) Accession Code Number - This must be integer or floating point -- i.e., not alphanumeric -- and must be present. It may be simply an integer which decodes to a null message.

(vii) Remarks - Up to five lines of text, each line enclosed in quotation marks, with the final line terminated in an exclamation mark. If no remarks are to be entered, this field consists of the symbols "!".

Experience has indicated that these data format restrictions are not severe and can be easily interpreted by secretarial or keypunching personnel. Figure 1 shows how a typical bibliographic entry is arranged.

(d) Search Requirements: The restrictions placed upon the capability of the program to search and retrieve are also somewhat arbitrary and are compromise solutions which attempt to balance the utility of the program against its cost as reckoned in terms of computing time and core requirements.

Three types of search are possible: Author, Key Expression, and Code Group.

ACOUST. A98 0061

ANERT, V.
(M.V. LOMONOSOV MOSCOW STATE UNIV., USSR)

ABSOLUTE MAXIMUM OF THE FREQUENCY RESPONSE CURVE FOR ACOUSTIC FEEDBACK

AKUST. ZH. (USSR)
V/I: 19(1), P: 1-8 , JAN.-FEB. 1973 , R: 14

IT IS SHOWN THAT THE SOUND AMPLIFICATION IN ENCLOSURES CAN BE CALCULATED ON THE BASIS OF THE STATISTICAL THEORY OF FREQUENCY RESPONSE CURVES. THE INFLUENCE OF THE ABSOLUTE MAXIMA OF THE FREQUENCY CURVES FOR AN ENCLOSURE ON THE ACOUSTIC GAIN PROVIDED BY A REAL SYSTEM IS DETERMINED FROM AN ANALYSIS OF EXISTING DATA

JULY-AUG. 1973 PHYS. ACOUST. (USA)

AN A59483 P 0583 EN 08 TW 000 WT 000 S P1973 TP ARTC L ENG

Figure 1a. A Typical Bibliographic Retrieval from the CAN/SDI System.

"ANERT, V. (M.V. LOMONOSOV MOSCOW STATE UNIV., USSR)"
"ABSOLUTE MAXIMUM OF THE FREQUENCY RESPONSE CURVE FOR ACOUSTIC"
"FEEDBACK!"
"AKUST. ZH. (USSR) 19, 1, 1-8, 1973!"
"IT IS SHOWN THAT THE SOUND AMPLIFICATION IN ENCLOSURES CAN BE"
"CALCULATED ON THE BASIS OF THE STATISTICAL THEORY OF FREQUENCY"
"RESPONSE CURVES. THE INFLUENCE OF THE ABSOLUTE MAXIMA OF THE"
"FREQUENCY CURVES FOR AN ENCLOSURE ON THE ACOUSTIC GAIN PROVIDED"
"BY A REAL SYSTEM IS DETERMINED FROM AN ANALYSIS OF EXISTING"
"DATA!"
0
0
"TRANSLATION IN JULY-AUG ISSUE OF PHYS. ACOUST. (USA)!"

Figure 1b. The Format of the Above Item Coded for Program INFORM.

During an Author search, only the Author field of the data file entries is scanned and only one author's name is sought, although this may occur in any position within the field. Thus an Author search for SMITH, A.A. will retrieve all the publications for which A.A. Smith is author or co-author. Note that when the Author field is used to indicate also the author's institutional affiliation, an Author search can, in effect, be an institution search. Thus, for example, an Author search could retrieve publications authored by employees of the Atmospheric Environment Service if the appropriate information had been incorporated into the author field in the data file.

A Code Group search scans only the subject code numbers and seeks a match with the search group, taking into account that zeros are wild. Thus, if 2 as the first code number represents ACOUSTICS and 1 as the last number decodes as INSTRUMENTS, then a search for 2, 0, 0, 1 would retrieve all papers dealing with acoustic instrumentation while a search for 2, 0, 0, 0 would return all papers on acoustics.

It may be apparent at this juncture that the Author and Code Group searches are apt to be fairly fast but of limited use. The Author search may often be avoided, for example, if one has taken the trouble to arrange the items in the data file alphabetically by author and if one has a page copy of the data file. The Code Group search is a specialist's search, useful when the user is seeking information on a well-defined topic but only when that topic has been included in the coding scheme and all the papers in the bibliography have been coded.

The most complicated, most flexible, most expensive, and potentially most useful type of search is the Key Expression search during which both the Title field and the Abstract field are scanned for the occurrence or absence of one or more specified expressions. Here an expression may be a phrase, a word, a word fragment, or essentially any group of alphanumeric symbols. Up to 4 such expressions can be sought simultaneously. Provision is made for adding "but not" terms by simply prefacing an expression with the symbol /. Thus a Key Expression search for the two expressions WAVE and /OCEAN would retrieve all those papers for which the word WAVE occurred in either the title or the abstract except those for which the word OCEAN was found in either the title or the abstract. Clearly the Key Expression search is a powerful and flexible method of retrieving information from a data file.

Adequate user instructions are provided by the program during execution. Examples of search techniques and typical results are given in Chapter 3 of this report.

In its present form, the program can scan a data file of up to 5000 items if the available computer room is sufficient to accommodate a file of this size. Up to 100 items can be retrieved at a time. If this number is exceeded, the search stops and a warning message is printed. Many successive searches can be made without re-initialization of the program. After each search, the user has the option of printing the results either on the time-sharing terminal or on the computer disk area to be fed to a line printer at a later time.

3. Examples of the Use of the Program

The following pages show the terminal printout corresponding to a complete run of the program INFORM. Successful Author, Key Expression, and Code Group searches are made and the retrieved data are printed on the terminal and on the disk. Also shown are an example of an unsuccessful Author search and one case in which BASIC responds with a query when given an incorrect command.

The circled numbers on the printout refer to notes which follow. The system commands shown are those used with a DEC PDP-10 computer operated by Dataline Systems Ltd. of Toronto. The prospective user should consult his local computing consultant if he is in doubt with regard to the systems operation of the computer he is to use.

Acknowledgements

The author is grateful to Mr. G.H. Gilbert for his interest and encouragement throughout the course of development of this program and for his assistance in the preparation of this report.

The author also wishes to thank Miss Janet MacDougall for reviewing the manuscript and Dr. J.W.S. Young of the Air Quality Research Division of AES for helpful discussions concerning the design criteria of the program.

Dr. Young has written an information retrieval program in the FORTRAN language to perform search operations in a manner similar to that described in this report. Tests have shown that neither program has a significant advantage over the other in terms of computing time required, so that the decision to use a particular language remains dependent largely on non-economic factors.

APPROVED



J.R.H. Noble,
Assistant Deputy Minister
Atmospheric Environment Service.

```
LOG
JOB 9    DATALINE SYSTEM 2 DSL2B3
NOTICE.TXT LAST CHANGED  11:27  25-SEP-73
#357/16
PASSWORD:
PROJECT?REAMJC
  10:01:34        14-OCT-73        TTY111

.MOUNT D1554
D1554 REQUEST ACCEPTED
*10:03 D1554 MOUNTED ON DTA1
*^C

.R PIP

*DSK:/X_DTA1:INFORM.BAS,ALIB.DAT

*^C

.R BASIC

NEW OR OLD--OLD
OLD FILE NAME--INFORM

READY
RUN

INFORM          10:04          14-OCT-73

PROGRAM INFORM; UPDATED 12-OCT-73

THIS PROGRAM PERMITS AN INFORMATION RETRIEVAL SEARCH OF YOUR
PERSONAL DATA FILE BY AUTHOR, BY KEY WORD(S) OR PHRASE(S), OR
BY CODE NUMBER.

PLEASE ENCLOSE SEARCH EXPRESSIONS IN DOUBLE QUOTES.

DO YOU REQUIRE FURTHER INFORMATION   ?NO

WHAT IS THE NAME OF YOUR DATA FILE
(ENCLOSE IN DOUBLE QUOTES)     ?"ALIB.DAT"

IS THIS SEARCH TO BE FOR AUTHOR'S NAME, FOR KEY WORD(S)
OR PHRASE(S), OR FOR A CODE GROUP  (REPLY A, K, OR C)     ?A
```

① ② ③ ④ ⑤ ⑥ ⑦

```
ENTER AUTHOR'S NAME (ENCLOSE IN DOUBLE QUOTES)      ?"STEWART, R.W."

DO YOU WISH TO IMPOSE RESTRICTIONS ON THE DATE OF PUBLICATION     ?NO
```

(7)

```
THERE IS  1  ITEM IN THE LIST.
DO YOU WANT IT PRINTED     ?YES

DO YOU WANT THE ABSTRACTS PRINTED     ?YES
ON THE DISK OR ON THE TELETYPE?  (REPLY WITH D OR T)     ?T
```

(8)

```
1. STEWART, R.W.  (INST. OF OCEANOGRAPHY, U.B.C.)
   TURBULENCE AND WAVES IN A STRATIFIED ATMOSPHERE
   RADIO SCIENCE, 4, 12, 1269-1278, DECEMBER 1969
   TO DISTINGUISH RIGOUROUSLY BETWEEN WAVE MOTION AND TURBULENCE IN A
   STRATIFIED FLUID SEEMS IMPOSSIBLE, ALTHOUGH USEFUL APPROXIMATIONS
   SEEM FEASIBLE. THE TASK IS MADE MORE DIFFICULT BECAUSE PROPERTIES OF
   TURBULENCE ARE NOT LIKE THOSE DESCRIBED IN MOST THEORIES OF TURBULENCE
   EVEN WHEN THE REYNOLDS NUMBER IS HIGH.
  1  2  1  4 ,
     THIS ITEM IS NOT IN YOUR FILES
  MGA 21.10-224
```

```
DO YOU WANT TO MAKE ANOTHER SEARCH      ?YES
SAME DATA FILE     ?YES
```

(9)

```
IS THIS SEARCH TO BE FOR AUTHOR'S NAME, FOR KEY WORD(S)
OR PHRASE(S), OR FOR A CODE GROUP  (REPLY A, K, OR C)     ?K


ENCLOSE EACH EXPRESSION IN DOUBLE QUOTES.
TERMINATE FINAL EXPRESSION WITH  !
PREFACE AN EXPRESSION WITH  /  TO INDICATE A 'BUT NOT' FUNCTION:
 KEY EXPRESSION  1      ?INTERNAL
 KEY EXPRESSION  2      ?GRAVITY
 KEY EXPRESSION  3      ?WAVE
 KEY EXPRESSION  4      ?/MODES!

DO YOU WISH TO IMPOSE RESTRICTIONS ON THE DATE OF PUBLICATION     ?NO
```

(10)

```
THERE ARE  2  ITEMS IN THE LIST.
DO YOU WANT THEM PRINTED     ?YES

DO YOU WANT THE ABSTRACTS PRINTED     ?YES
ON THE DISK OR ON THE TELETYPE?  (REPLY WITH D OR T)     ?D

YOUR DATA HAVE BEEN WRITTEN IN A DISK FILE NAMED ALIB1.OPT

DO YOU WANT TO MAKE ANOTHER SEARCH      ?YES
```

(11)

```
SAME DATA FILE    ?YES

IS THIS SEARCH TO BE FOR AUTHOR'S NAME, FOR KEY WORD(S)
OR PHRASE(S), OR FOR A CODE GROUP  (REPLY A, K, OR C)    ?C
ENTER FOUR-NUMBER CODE GROUP. REMEMBER THAT ZERO IS WILD AND
SEPARATE NUMBERS WITH COMMAS    ?3,0,0,0

DO YOU WISH TO IMPOSE RESTRICTIONS ON THE DATE OF PUBLICATION    ?NO
```

(12)

```
THERE IS  1  ITEM IN THE LIST.
DO YOU WANT IT PRINTED    ?YES

DO YOU WANT THE ABSTRACTS PRINTED    ?NO
ON THE DISK OR ON THE TELETYPE?  (REPLY WITH D OR T)    ?D

YOUR DATA HAVE BEEN WRITTEN IN A DISK FILE NAMED ALIB2.OPT

DO YOU WANT TO MAKE ANOTHER SEARCH    ?YES
SAME DATA FILE    ?YES
```

(13)

```
IS THIS SEARCH TO BE FOR AUTHOR'S NAME, FOR KEY WORD(S)
OR PHRASE(S), OR FOR A CODE GROUP  (REPLY A, K, OR C)    ?A

ENTER AUTHOR'S NAME (ENCLOSE IN DOUBLE QUOTES)    ?"TRUDEAU, P.E."

DO YOU WISH TO IMPOSE RESTRICTIONS ON THE DATE OF PUBLICATION    ?NO


SEARCH UNSUCCESSFUL.
DO YOU WANT TO MAKE ANOTHER SEARCH    ?NO
```

(14)

```
RUN TIME:  25.92 SECS.

READY
SSS

? WHAT?
READY
SYS

EXIT
^C
```

(15)

```
.DIR

DIRECTORY        357,16  10:14    14-OCT-73

INFORM  BAS        23     <457>   14-OCT-73
ALIB    DAT        12     <057>   27-SEP-73
ALIB1   OPT        01     <457>   14-OCT-73
ALIB2   OPT        01     <457>   14-OCT-73

TOTAL BLOCKS       37
```

(16)

```
.TYPE ALIB1.OPT
  1. CURRY, M.J.
     INTERNAL GRAVITY WAVES OF TROPOSPHERIC ORIGIN
```

PH.D. THESIS, UNIVERSITY OF WESTERN ONTARIO, 1973
* ABSTRACT HAS NOT BEEN ENTERED *
1 4 1 0 , 1 4 2 2 , 1 4 3 1 , 1 4 8 0 , 4 4 7 8 ,
BOOK: PRIVATE LIBRARY M.J.C.

2. TOLSTOY, I. AND PAN, P.
SIMPLIFIED ATMOSPHERIC MODELS AND THE PROPERTIES OF LONG PERIOD INTERNAL AND SURFACE GRAVITY WAVES
J. ATMOSPHERIC SCIENCES, 27, 1, 31-51, 1970
* ABSTRACT HAS NOT BEEN ENTERED *
4 19 7 0 , 1 4 4 0 ,
THIS ITEM IS NOT IN YOUR FILES
MGA 21.6-271

(17)

.PRINT ALIB2.OP1
FILES PRINTED:
ALIB2 .OP1

.DISMOUNT D1A1
D1A1 DISMOUNT REQUESTED
*10:16 D1554 DISMOUNTED
*^C

(18)

.K
CONFIRM: E
JOB 9, USER [357,16] LOGGED OFF TTY111SYSTEM 2 10:17:06 14-OCT-73
SAVED ALL FILES
RUNTIME 0 MIN, 36.19 SEC

Notes on The Operation of the Program

1. This is the login procedure by which access to the computer is gained.

2. Since both the program and the data file are stored on magnetic tape, it is necessary to ask the computer operator to mount the tape on a tape drive so that it can be read.

3. This version of BASIC and hence this program require that the program and data files both be stored on disk. The file transfer program PIP is used to copy the files from the tape to the disk.

4. R BASIC is the command which prepares the computer for operation in the BASIC language. The user responds to the question "NEW OR OLD --" by typing "OLD" to signify that he is using a program which already exists rather than creating a new one. INFORM is, of course, the name given to the program. When the computer had found the program it prints READY. The user then starts the program by typing RUN.

5. This is the introductory heading printed by INFORM itself. Since the user replies NO to the question of the necessity for further information, the more detailed text is not printed.

6. Here, INFORM asks what data file is to be searched. The user responds with the name of the file, enclosed in quotation marks. (A copy of this data file is given in Appendix B to this report).

7. The user instructs the computer to perform an Author search for the publications of R.W. Stewart, regardless of their publication date.

8. The search completed, the computer advises that it has found one such paper and asks for printing instructions. Since the retrieval list is a short one, the user indicates that he wishes to have it printed on the time-sharing terminal.

9. When the printout is completed, the computer asks if another search is required and if that search is to be done on the same data file.

10. The computer is instructed to perform a Key Expression search for those papers for which the words INTERNAL, GRAVITY, and WAVE appear in either the title or the abstract but to exclude those which mention MODES. *

11. The computer reports that two papers have been retrieved and asks for printing instructions. The user elects to have the output written on the disk. The computer advises that the output file has been named ALIB1.OPT. The user then chooses to conduct another search.

12. A Code Group search is requested for all those papers having 3 as the first number in any code group. (Recall that zeros are wild). An inspection of the data file (Appendix B) shows that there is only one paper so coded -- a paper by C.O. Hines which has 3, 7, 1, 8 as its second code group.

13. The computer signifies that it has found the required paper. The user decides to have this output also written on the disk and to conduct another search.

14. The computer searches unsuccessfully for the scientific writings of P.E. Trudeau. The user decides that no further searches are to be made.

15. The computer prints READY to signify that it is prepared to accept another command. The user mistakenly types SSS. The computer cannot recognize this command and ignores it. The user then types SYS to return to the system -- i.e., to exit from BASIC.

16. The user types DIR to obtain a directory listing of his disk area. Note that the output files ALIB1.OPT and ALIB2.OPT have indeed been created.

17. The user decides to have ALIB1.OPT printed on the time-sharing terminal and to have ALIB2.OPT printed on the line printer.

18. Finally, the user requests that the operator dismount the tape and logs out. The entire operation has taken 16 minutes of elapsed time and 36 seconds of computer time.

---

* Note that the user has omitted the quotation marks. This is possible in this case because the input expressions do not contain blanks or delimiters. If in doubt, the user should enclose each expression in quotation marks.

APPENDIX A

THE PROGRAM

This Appendix contains a listing of the program itself, called INFORM in the present application, followed by an interpretation of the program based on command instructions to the computer.

```
10      MARGIN 80
20      MARGIN ALL 80
30      PRINT <PA>
39      LET S8$="      "
40      LET S9$= "REPLY WITH YES OR NO"
50      PRINT "  PROGRAM INFORM; UPDATED 12-OCT-73"
60      PRINT
70      PRINT "  THIS PROGRAM PERMITS AN INFORMATION RETRIEVAL SEARCH OF YOUR"
80      PRINT "  PERSONAL DATA FILE BY AUTHOR, BY KEY WORD(S) OR PHRASE(S), OR"
90      PRINT "  BY CODE NUMBER."
100     PRINT
110     PRINT "  PLEASE ENCLOSE SEARCH EXPRESSIONS IN DOUBLE QUOTES."
120     PRINT
150     PRINT "  DO YOU REQUIRE FURTHER INFORMATION  ";
160     INPUT Q$
170     IF Q$="YES" THEN 210
180     IF Q$="NO" THEN 626
190     PRINT S9$
200     GO TO 150
210     PRINT <PA>
220     PRINT "  ORGANIZATION OF THE DATA:"
230     PRINT
240     PRINT "  THE INFORMATION IN THE DATA FILE CONSISTS OF BIBLIOGRAPHIC"
250     PRINT "  INFORMATION AND IS TO BE ENTERED IN THE FORM"
260     PRINT "                    AUTHOR"
270     PRINT "                    TITLE"
280     PRINT "                    REFERENCE"
290     PRINT "                    ABSTRACT"
300     PRINT "                    CODE NUMBERS"
310     PRINT "                    ACCESSION NUMBER"
320     PRINT "                    REMARKS"
330     PRINT "  WHERE"
340     PRINT " 1.  AUTHOR IS A LIST OF AUTHORS' NAMES, IN THE FORM"
350     PRINT "              SURNAME, INITIALS"
360     PRINT "       EG:  SMITH, A.B. AND JONES, C.D."
370     PRINT "       RESTRICT AUTHOR TO A SINGLE LINE OF TEXT AND ENCLOSE IN"
380     PRINT "       DOUBLE QUOTES."
390     PRINT " 2.  TITLE REPRESENTS UP TO 5 LINES OF TEXT, EACH LINE ENCLOSED"
400     PRINT "       IN DOUBLE QUOTES, WITH THE FINAL LINE TERMINATED WITH AN"
410     PRINT "       EXCLAMATION MARK."
420     PRINT " 3.  REFERENCE REPRESENTS UP TO 5 LINES OF TEXT, EACH LINE"
430     PRINT "       ENCLOSED IN DOUBLE QUOTES. THE LAST FIVE CHARACTERS MUST"
440     PRINT "       BE THE YEAR OF PUBLICATION FOLLOWED BY AN EXCLAMATION"
450     PRINT "       MARK."
460     PRINT " 4.  ABSTRACT REPRESENTS UP TO 20 LINES OF TEXT, EACH LINE"
470     PRINT "       ENCLOSED IN DOUBLE QUOTES WITH THE FINAL LINE TERMINATED"
480     PRINT "       WITH AN EXCLAMATION MARK. IF NO ABSTRACT IS AVAILABLE,"
490     PRINT "       SIMPLY ENTER AN EXCLAMATION MARK ENCLOSED IN DOUBLE"
500     PRINT "       QUOTES."
510     PRINT " 5.  CODE NUMBERS ARE REFERENCE GROUPS OF 4 INTEGERS EACH, TO"
520     PRINT "       A MAXIMUM OF 5 GROUPS. THE FINAL GROUP IS FOLLOWED BY A"
530     PRINT "       SINGLE ZERO. ZEROS WITHIN THE CODE GROUPS ARE WILD, WITH"
540     PRINT "       THE EXCEPTION THAT ZERO IS NOT ALLOWED AS THE FIRST"
550     PRINT "       INTEGER IN A GROUP."
560     PRINT "            EG:   1, 0, 4, 2,   2,4,0,0   1,1,7,2,   0"
570     PRINT " 6.  ACCESSION NUMBER IS YOUR OWN FILE CODE NUMBER AND MUST BE"
580     PRINT "       INTEGER OR FLOATING POINT. A ZERO HERE INDICATES THAT YOUR"
590     PRINT "       FILE DOES NOT CONTAIN THIS ITEM."
```

```
600     PRINT " 7.  REMARKS REPRESENTS UP TO 5 LINES OF TEXT, EACH LINE"
610     PRINT "     ENCLOSED IN DOUBLE QUOTES WITH THE FINAL LINE TERMINATED"
620     PRINT "     WITH AN EXCLAMATION MARK."
626     DIM W$(10)
627     MAT READ W$
630     PRINT <PA>
631     DIM A1$(100),A2$(500),A3$(500),A4$(1000),A5(1200)
632     DIM A6(100),A7$(500),A8$(4),A9(4),S2(4)
633     DIM H2(100),H3(100),H4(100),H5(100),H6(100)
634     DEF FNA(I,J)= J+(I-1)*5
635     DEF FNB(I,J)= J+(I-1)*20
636     DEF FNC(I,J,K)= K+(J-1)*4+(I-1)*24
637     DEF FND(J,K)= K+(J-1)*4
638     LET L2=0
639     PRINT "  WHAT IS THE NAME OF YOUR DATA FILE"
640     PRINT "  (ENCLOSE IN DOUBLE QUOTES)    ";
641     INPUT F1$
642     FILE #1, F1$
643     GO TO 650
644     PRINT "  SAME DATA FILE    ";
645     INPUT Q$
646     IF Q$="YES" THEN 650
647     IF Q$="NO" THEN 639
648     PRINT S9$
649     GO TO 640
650     DIM Z2$(5),Z3$(5),Z4$(20),Z5(25),Z7$(5)
660     PRINT
661     PRINT "  IS THIS SEARCH TO BE FOR AUTHOR'S NAME, FOR KEY WORD(S)"
670     PRINT "  OR PHRASE(S), OR FOR A CODE GROUP  (REPLY A, K, OR C)    ";
680     INPUT Q1$
690     IF Q1$="A" THEN 730
700     IF Q1$="K" THEN 770
710     IF Q1$="C" THEN 870
720     GO TO 650
730     PRINT
731     PRINT "  ENTER AUTHOR'S NAME (ENCLOSE IN DOUBLE QUOTES)    ";
740     INPUT T1$
750     LET S1=LEN(T1$)
760     GO TO 949
770     PRINT
771     PRINT
772     PRINT "  ENCLOSE EACH EXPRESSION IN DOUBLE QUOTES."
780     PRINT "  TERMINATE FINAL EXPRESSION WITH  ! "
785     PRINT "  PREFACE AN EXPRESSION WITH  /  TO INDICATE A";
786     PRINT " 'BUT NOT' FUNCTION:"
787     LET N7=0
790     FOR I=1 TO 4
800         PRINT "  KEY EXPRESSION ";I;"    ";
810         INPUT A8$(I)
820         LET S2(I)=LEN(A8$(I))
825         IF LEFT$(A8$(1),1)="/" THEN 827
826         GO TO 830
827         LET N7=N7+1
830         LET N1=1
840         IF RIGHT$(A8$(I),1)="!" THEN 938
850         NEXT I
860     GO TO 937
870     PRINT "  ENTER FOUR-NUMBER CODE GROUP. REMEMBER THAT ZERO IS WILD AND"
880     PRINT "  SEPARATE NUMBERS WITH COMMAS    ";
```

```
890     MAT INPUT A9
900     IF A9(1)=0 THEN 920
910     GO TO 940
920     PRINT "  *** ZERO IS ILLEGAL AS FIRST CODE NUMBER ***"
930     GO TO 870
937     IF RIGHT$(A8$(N1),1)<>"!" THEN 940
938     LET S2(N1)=S2(N1)-1
939     LET A8$(N1)=LEFT$(A8$(N1),S2(N1))
940     IF N7=N1 THEN 942
941     GO TO 949
942     PRINT
943     PRINT "  ** YOU MUST CHOOSE AT LEAST ONE EXPRESSION WHICH"
944     PRINT "  IS NOT A 'BUT NOT' PHRASE.  TRY AGAIN. **"
945     PRINT
946     GO TO 770
949     PRINT
950     PRINT "  DO YOU WISH TO IMPOSE RESTRICTIONS ON THE DATE OF";
960     PRINT " PUBLICATION   ";
970     INPUT Q2$
980     IF Q2$="YES" THEN 1020
990     IF Q2$="NO" THEN 1070
1000    PRINT S9$
1010    GO TO 950
1020    PRINT "  EARLIEST YEAR   ";
1030    INPUT Y1
1040    PRINT "  LATEST YEAR   ";
1050    INPUT Y2
1060    GO TO 1090
1070    LET Y1=1
1080    LET Y2=2001
1090    LET L0=0
1110    RESTORE #1
1120    FOR I=1 TO 5000
1130        IF END #1 THEN 2100
1135        LET N2=N3=N4=N5=N6=0
1140        INPUT #1, Z1$
1150        FOR J1=1 TO 5
1160            INPUT #1, Z2$(J1)
1170            LET N2=J1
1180            IF RIGHT$(Z2$(J1),1)="!" THEN 1195
1190            NEXT J1
1195        LET P1=LEN(Z2$(N2))
1196        LET Z2$(N2)=LEFT$(Z2$(N2),P1-1)
1200        FOR J1=1 TO 5
1210            INPUT #1, Z3$(J1)
1220            LET N3=J1
1230            IF RIGHT$(Z3$(J1),1)="!" THEN 1245
1240            NEXT J1
1245        LET P1=LEN(Z3$(N3))
1246        LET Z3$(N3)=LEFT$(Z3$(N3),P1-1)
1250        FOR J1=1 TO 20
1260            INPUT #1, Z4$(J1)
1270            LET N4=J1
1280            IF RIGHT$(Z4$(J1),1)="!" THEN 1295
1290            NEXT J1
1295        LET P1=LEN(Z4$(N4))-1
1296        IF P1=0 THEN 1300
1297        LET Z4$(N4)=LEFT$(Z4$(N4),P1)
1300        FOR J2=1 TO 6
```

```
1310            INPUT #1, Z5(FND(J2,1))
1320            IF Z5(FND(J2,1))=0 THEN 1410
1330            LET N5=J2
1340            FOR J3=2 TO 4
1350                INPUT #1, Z5(FND(J2,J3))
1360                NEXT J3
1370            NEXT J2
1410        INPUT #1, Z6
1420        FOR J1=1 TO 5
1430            INPUT #1, Z7$(J1)
1440            LET N6=J1
1450            IF RIGHT$(Z7$(J1),1)="!" THEN 1461
1460            NEXT J1
1461        IF Z7$(N6)="!" THEN 1463
1462        GO TO 1464
1463        LET Z7$(N6)=" "
1464        LET P1=LEN(Z7$(N6))-1
1465        IF P1=0 THEN 1470
1466        LET Z7$(N6)=LEFT$(Z7$(N6),P1)
1470        IF Q1$="K" THEN 1550
1480        IF Q1$="C" THEN 1760
1490        LET S3=LEN(Z1$)
1491        IF S3<S1 THEN 2090
1500        FOR J1=1 TO S3-S1+1
1510            LET X1$=MID$(Z1$,J1,S1)
1520            IF X1$=T1$ THEN 1850
1530            NEXT J1
1540        GO TO 2090
1550        LET M1=1
1555        IF M1=N1+1 THEN 2090
1556        LET S5=S2(M1)
1557        LET V$=A8$(M1)
1558        IF LEFT$(V$,1)="/" THEN 1560
1559        GO TO 1562
1560        LET S5=S5-1
1561        LET V$=RIGHT$(V$,S5)
1562            FOR M2=1 TO N2
1570                LET S4=LEN(Z2$(M2))
1575                IF S4<S5 THEN 1620
1580                FOR J3=1 TO S4-S5+1
1590                    LET X1$=MID$(Z2$(M2),J3,S5)
1591                    IF LEFT$(A8$(M1),1)="/" THEN 1593
1592                    GO TO 1600
1593                    IF X1$=V$ THEN 2090
1594                    GO TO 1610
1600                    IF X1$=V$ THEN 1710
1610                    NEXT J3
1620                NEXT M2
1630            FOR J4=1 TO N4
1640                LET S4=LEN(Z4$(J4))
1645                IF S4<S5 THEN 1690
1650                FOR J3=1 TO S4-S5+1
1660                    LET X1$=MID$(Z4$(J4),J3,S5)
1661                    IF LEFT$(A8$(M1),1)="/" THEN 1663
1662                    GO TO 1670
1663                    IF X1$=V$ THEN 2090
1664                    GO TO 1680
1670                    IF X1$=V$ THEN 1710
1680                    NEXT J3
```

```
1690            NEXT J4
1695        IF LEFT$(A8$(M1),1)="/" THEN 1710
1700        GO TO 2090
1710        IF M1=N1 THEN 1850
1720        LET M1=M1+1
1730        GO TO 1555
1760        FOR J1=1 TO N5
1770            IF Z5(FND(J1,1))=0 THEN 2090
1780            FOR J2=1 TO 4
1790                IF A9(J2)=0 THEN 1810
1800                IF A9(J2)<>Z5(FND(J1,J2)) THEN 1830
1810                NEXT J2
1820            GO TO 1850
1830            NEXT J1
1840        GO TO 2090
1850        LET X1$=RIGHT$(Z3$(N3),4)
1870        LET K=VAL(X1$)
1880        IF K<Y1 THEN 2090
1890        IF K>Y2 THEN 2090
1900        LET L0=L0+1
1910        LET A1$(L0)=Z1$
1915        LET H2(L0)=N2
1920        FOR J1=1 TO N2
1930            LET A2$(FNA(L0,J1))=Z2$(J1)
1940            NEXT J1
1945        LET H3(L0)=N3
1950        FOR J1=1 TO N3
1960            LET A3$(FNA(L0,J1))=Z3$(J1)
1970            NEXT J1
1975        LET H4(L0)=N4
1980        FOR J1=1 TO N4
1990            LET A4$(FNB(L0,J1))=Z4$(J1)
2000            NEXT J1
2005        LET H5(L0)=N5
2010        FOR J1=1 TO N5
2020            FOR J2=1 TO 4
2030                LET A5(FNC(L0,J1,J2))=Z5(FND(J1,J2))
2040            NEXT J2,J1
2050        LET H6(L0)=N6
2055        LET A6(L0)=Z6
2060        FOR J1=1 TO N6
2070            LET A7$(FNA(L0,J1))=Z7$(J1)
2080            NEXT J1
2085    IF L0=100 THEN 2201
2090        NEXT I
2100    PRINT
2110    PRINT
2120    IF L0=0 THEN 2140
2130    GO TO 2206
2140    PRINT "  SEARCH UNSUCCESSFUL."
2150    PRINT "  DO YOU WANT TO MAKE ANOTHER SEARCH    ";
2160    INPUT Q$
2170    IF Q$="YES" THEN 640
2180    IF Q$="NO" THEN 3310
2190    PRINT S9$
2200    GO TO 2150
2201    PRINT
2202    PRINT "  ** WARNING: YOU HAVE EXCEEDED SYSTEM RETRIEVAL"
2203    PRINT  "  CAPACITY. USE A MORE SPECIFIC SEARCH EXPRESSION"
```

```
2204    PRINT
2205    PRINT
2206    IF L0=1 THEN 2208
2207    GO TO 2215
2208    PRINT "  THERE IS ";L0;" ITEM IN THE LIST."
2209    PRINT "  DO YOU WANT IT PRINTED   ";
2210    GO TO 2230
2215    PRINT "  THERE ARE ";L0;" ITEMS IN THE LIST."
2220    PRINT "  DO YOU WANT THEM PRINTED   ";
2230    INPUT Q$
2240    IF Q$="YES" THEN 2280
2250    IF Q$="NO" THEN 2150
2260    PRINT S9$
2270    GO TO 2210
2280    PRINT
2281    PRINT "  DO YOU WANT THE ABSTRACTS PRINTED   ";
2290    INPUT Q3$
2300    IF Q3$="YES" THEN 2340
2310    IF Q3$="NO" THEN 2340
2320    PRINT S9$
2330    GO TO 2280
2340    PRINT "  ON THE DISK OR ON THE TELETYPE?";
2350    PRINT "  (REPLY WITH D OR T)   ";
2360    INPUT Q4$
2370    IF Q4$="D" THEN 2760
2380    IF Q4$="T" THEN 2400
2390    GO TO 2350
2400    PRINT <PA>
2420    FOR I=1 TO L0
2430        PRINT USING "###. ", I;
2431        PRINT A1$(I)
2440        FOR J1=1 TO H2(I)
2450            PRINT S8$;A2$(FNA(I,J1))
2460            NEXT J1
2470        FOR J1=1 TO H3(I)
2480            PRINT S8$;A3$(FNA(I,J1))
2490            NEXT J1
2500        IF Q3$="NO" THEN 2570
2510        IF A4$(FNB(I,1))="!" THEN 2560
2520        FOR J1=1 TO H4(I)
2530            PRINT S8$;A4$(FNB(I,J1))
2540            NEXT J1
2550        GO TO 2570
2560        PRINT "      * ABSTRACT HAS NOT BEEN ENTERED *"
2570        IF H5(I)=0 THEN 2650
2575        PRINT "  ";
2580        FOR J1=1 TO H5(I)
2590            PRINT " ";A5(FNC(I,J1,1));
2600            FOR J2=2 TO 4
2610                PRINT A5(FNC(I,J1,J2));
2615                NEXT J2
2616            PRINT ",";
2620            NEXT J1
2630        PRINT
2650        IF A6(I)<10 THEN 2680
2660        PRINT S8$;"  FILE ACCESSION NUMBER  ";A6(I)
2670        GO TO 2690
2680        LET F=A6(I)+1
2681        IF W$(F)="NULL MESSAGE" THEN 2690
```

```
2682        PRINT S8$;W$(F)
2690        FOR J1=1 TO H6(I)
2700            PRINT S8$;A7$(FNA(I,J1))
2710            NEXT J1
2720        PRINT
2730        PRINT
2740        NEXT I
2750    GO TO 2150
2760    LET L2=L2+1
2761    LET L1=2
2762    LET B1$= "ALIB"+STR$(L2)+".OPT"
2763    FILE #L1, B1$
2770    SCRATCH #L1
2790    FOR I=1 TO L0
2800        PRINT USING #L1, "###. ",I;
2801        PRINT #L1, A1$(I)
2810        FOR J1=1 TO H2(I)
2820            PRINT #L1, S8$;A2$(FNA(I,J1))
2830            NEXT J1
2840        FOR J1=1 TO H3(I)
2850            PRINT #L1, S8$;A3$(FNA(I,J1))
2860            NEXT J1
2870        IF Q3$="NO" THEN 2940
2880        IF A4$(FNB(I,1))="!" THEN 2930
2890        FOR J1=1 TO H4(I)
2900            PRINT #L1, S8$;A4$(FNB(I,J1))
2910            NEXT J1
2920        GO TO 2940
2930        PRINT #L1, "    * ABSTRACT HAS NOT BEEN ENTERED *"
2940        IF H5(I)=0 THEN 3020
2945        PRINT #L1, "  ";
2950        FOR J1=1 TO H5(I)
2960            PRINT #L1, " ";A5(FNC(I,J1,1));
2970            FOR J2=2 TO 4
2980                PRINT #L1, A5(FNC(I,J1,J2));
2985                NEXT J2
2986            PRINT #L1, ",";
2990            NEXT J1
3015        PRINT #L1
3020        IF A6(I)<10 THEN 3050
3030        PRINT #L1, S8$;"FILE ACCESSION NUMBER ";A6(I)
3040        GO TO 3060
3050        LET F=A6(I)+1
3051        IF W$(F)="NULL MESSAGE" THEN 3060
3052        PRINT #L1, S8$;W$(F)
3060        FOR J1=1 TO H6(I)
3070            PRINT #L1, S8$;A7$(FNA(I,J1))
3080            NEXT J1
3090        PRINT #L1
3100        PRINT #L1
3110        NEXT I
3160    PRINT
3170    PRINT "  YOUR DATA HAVE BEEN WRITTEN IN A";
3180    PRINT " DISK FILE NAMED ";B1$
3190    PRINT
3200    GO TO 2150
3210    DATA "  THIS ITEM IS NOT IN YOUR FILES"
3220    DATA "  BOOK: PRIVATE LIBRARY M.J.C."
3230    DATA "  BOOK: SECTION LIBRARY AREA"
```

```
3240    DATA "  BOOK: LIBRARY AES HDQTS"
3250    DATA "  CURRENTLY HELD: NOT FILED"
3260    DATA "NULL MESSAGE"
3270    DATA "NULL MESSAGE"
3280    DATA "NULL MESSAGE"
3290    DATA "NULL MESSAGE"
3300    DATA "NULL MESSAGE"
3310    END
```

Interpretation of the Program

Lines 10 and 20 - Set the width of lines for all input and output to 80 characters instead of the default condition of 72 characters. (This may not be required for some systems).

Lines 50-210 - Print some introductory information and ask if further instructions are necessary.

Lines 220-620 - Print this message if required. (The text shown here merely represents a description of the data file. However, some 400 lines of text can be outputted to give the user a complete set of operating instructions should it prove necessary to provide such a feature).

Lines 626 and 627 - Read in the decoding information for the accession number location codes. This information is contained in DATA statements at the end of the program).

Lines 631-633 - Define indexing functions used to keep track of array locations. (These functions are necessary because BASIC does not permit the use of multi-dimensional arrays. Because it is convenient to think in terms of such arrays, they are simulated by using multiple indicies to address locations in a one-dimensional array).

Line 638 - Initialize a counter used in assigning disk file names.

Lines 639-649 - Establish the name of the data file to be searched.

Line 650 - Dimension the dummy variables through which the data will be read.

Lines 660-720 - Determine which type of search is to be performed.

Lines 730-760 - Input the data for an author search.

Lines 770-860 - Input the data for a key expression search. Determine whether the expressions are of the "but not" type. Count both the total number of expressions and the number of "but not" expressions. Exit upon encountering the symbol ! at the end of an expression.

Lines 870-930 - Input the data for a code group search and check for illegal use of zero.

Lines 937-939 - Check for the symbol ! at the end of the last key expression and remove it if necessary.

Lines 940-946 - Check to ensure that at least one of the search expressions is not a "but not" expression.

Lines 949-1080 - Establish limits on the date of publication. (Default limits: 1 and 2001).

Line 1090 - Initialize the counter used to enumerate the retrieval items.

Line 1110 - Initialize the data file for reading.

Lines 1120-1466 - Check for the end of the data file. Input the seven groups for the first data item. Remove the symbol ! from the end of each group where necessary. If the REMARKS groups is represented by "!", insert a blank.

Lines 1470-1480 - Select the appropriate set of search instructions.

Lines 1490-1540 - Author search: Scan author field. If the desired name is found, proceed to the date selection. Otherwise, go to the end of the loop in preparation for reading the next item in the data file.

Lines 1550-1700 - Key expression search: Scan the title and abstract fields. If the desired expression is found (or if a "but not" expression is not found) proceed to the next section. Otherwise, go to the end of the loop.

Lines 1710-1730 - Check to see if all the Key Expressions have been sought. If so, proceed to the date selection. If not, go back and search for another.

Lines 1760-1840 - Code Group Search: Scan for the desired group, remembering that zeros are wild. If it is found, proceed to the date selection. If not, go to the end of the loop.

Lines 1850-1890 - Date selection: If the date of publication falls within the allotted limits, proceed to the storage instructions. Otherwise, go to the end of the loop.

Lines 1900-2080 - Storage instructions: Increment the retrieval counter. Store all seven fields of the successfully retrieved item in the appropriate arrays, keeping track of the number of lines in each as an aid to printing.

Line 2085 - Check to see if the system retrieval arrays are full.

Line 2090 - End of Loop. Go back in preparation for reading the next item from the data file.

Lines 2120-2200 - If no papers have been retrieved, print this message and ask if another search is required.

Lines 2201-2203 - Print this message if 100 papers have been retrieved.

Lines 2206-2270 - Advise how many retrievals there have been and ask if a listing is desired. If so, continue. If not, go back and ask if another search is required.

Lines 2280-2390 - Ask for printing instructions.

Lines 2400-2740 - Print the list of retrieved items on the time-sharing terminal.

Line 2750 - Go back and ask whether another search is required.

Lines 2760-2770 - Open a disk file named ALIB1.OPT if this is the first disk printout, ALIB2.OPT if it is the second, and so on. Prepare the file for writing.

Lines 2790-3110 - Write the list of retrieved items into the appropriate disk file.

Lines 3160-3190 - Identify the output file used.

Line 3200 - Go back and ask whether another search is required.

Lines 3210-3300 - These messages are printed out for integer values of Accession code number from 0 to 9 respectively. (The phrase NULL MESSAGE is not printed but is used to reserve space for future location messages.

Appendix B

A SAMPLE DATA FILE

```
"ATKINSON, B.W."
"STRUCTURE OF THE THUNDER ATMOSPHERE, SOUTH-EAST ENGLAND, 1951-60!"
"WEATHER 22, 8, 335-345, 1967!"
"MEAN TEMPERATURE AND HUMIDITY PROFILES WERE CALCULATED USING RADIO-SONDE"
"DATA FROM CRAWLEY, OR LARKHILL, IN SOUTHERN ENGLAND, FOR EACH OCCASION"
"WHEN THUNDERSTORMS WERE OBSERVED WITHIN 100 MI OF THE RADIOSONDE STATION."
"THESE MEAN 'THUNDER ATMOSPHERE' PROFILES ARE PRESENTED AND COMPARED WITH"
"THE OVERALL MEAN PROFILES. THE DATA WERE DIVIDED ACCORDING TO SYNOPTIC"
"TYPE AND MEAN VALUES OF VARIOUS TEMPERATURE AND HUMIDITY PARAMETERS WERE"
"CALCULATED. SOME OF THESE VALUES ARE TABULATED HERE. STABILITY PATTERNS"
"ARE ALSO CONSIDERED AND THE MEAN PATTERN OF STABILITY FOR THUNDERSTORM"
"OCCASIONS IN THE DIFFERENT SEASONS AND FOR DIFFERENT WEATHER TYPES ARE"
"BRIEFLY DISCUSSED.  SUBJECT HEADINGS: 1. THUNDERSTORM ANALYSIS  .....!"
 4,4,4,8,  4,12,9,8,  4,13,9,8,  4,16,3,0,  0
9
"MGA 19.1-191!"
"REDDY, C.A.  (NCAR, BOULDER)"
"DUCTING OF INTERNAL GRAVITY WAVES IN A TEMPERATURE AND WIND STRATIFIED"
"ATMOSPHERE!"
"NATIONAL CENTER FOR ATMOSPHERIC RESEARCH, BOULDER, COLORADO, TECHNICAL"
"NOTES, NO. 43, SEPTEMBER 1969!"
"USING THE MULTILAYER APPROXIMATION FOR A REALISTIC ATMOSPHERE (IN THE"
"0-240 KM REGION) WITH HEIGHT VARIATIONS OF TEMPERATURE AND BACKGROUND"
"WINDS, THE IMPERFECTLY DUCTED MODES OF INTERNAL GRAVITY WAVES WITH"
"***** ARE IDENTIFIED. THE ANALYSIS BRINGS OUT CLEARLY THE IMPORTANCE OF"
"WIND INDUCED REFLECTIONS IN DETERMINING THE CHARACTERISTICS OF PARTIALLY"
"DUCTED INTERNAL GRAVITY WAVES. NEGLIGENCE OF WIND INDUCED REFLECTIONS"
"CAN LEAD TO SUBSTANTIAL ERRORS IN ANY QUANTITATIVE COMPARISON OF"
"THEORETICALLY PREDICTED AND OBSERVED GRAVITY WAVE MODES. EVEN IN THE"
"PRESENCE OF WIND EFFECTS, THE HORIZONTAL GROUP VELOCITY CANNOT EXCEED"
"THE HORIZONTAL PHASE VELOCITY FOR THESE PARTIALLY DUCTED MODES. ONE::::!"
 1,4,12,4,  1,4,1,0,  1,4,5,0,  0
4
"MGA 21.7-241!"
"STEWART, R.W.  (INST. OF OCEANOGRAPHY, U.B.C.)"
"TURBULENCE AND WAVES IN A STRATIFIED ATMOSPHERE!"
"RADIO SCIENCE, 4, 12, 1269-1278, DECEMBER 1969!"
"TO DISTINGUISH RIGOUROUSLY BETWEEN WAVE MOTION AND TURBULENCE IN A"
"STRATIFIED FLUID SEEMS IMPOSSIBLE, ALTHOUGH USEFUL APPROXIMATIONS"
"SEEM FEASIBLE. THE TASK IS MADE MORE DIFFICULT BECAUSE PROPERTIES  OF"
"TURBULENCE ARE NOT LIKE THOSE DESCRIBED IN MOST THEORIES OF TURBULENCE"
"EVEN WHEN THE REYNOLDS NUMBER IS HIGH.!"
 1,2,1,4,  0
0
"MGA 21.10-224!"
"TAKEDA, T.  (MCGILL UNIV. STORMY WEATHER GROUP)"
"NUMERICAL SIMULATION OF LARGE CONVECTIVE CLOUDS!"
"MCGILL UNIVERSITY, STORMY WEATHER GROUP, SCIENTIFIC REPORT MW-64,"
"DECEMBER 1969!"
"!"
 4,18,4,6,  4,17,1,5,  0
0
"MGA 21.10-375!"
"HINES, C.O."
"ATMOSPHERIC GRAVITY WAVES IN OUTLINE!"
"IN: WEBB, W.L. (ED) 'THERMOSPHERIC CIRCULATION'.  PROGRESS IN"
"ASTRONAUTICS AND AERONAUTICS, VOL. 27.  MIT PRESS, CAMBRIDGE, 1972!"
"!"
```

```
  1,4,0,0,  0
0
"MGA 23.10-232       ON ORDER, U.W.O. LIBRARY!"
"CURRY, M.J."
"INTERNAL GRAVITY WAVES OF TROPOSPHERIC ORIGIN!"
"PH.D. THESIS, UNIVERSITY OF WESTERN ONTARIO, 1973!"
"!"
  1,4,1,0,  1,4,2,2,  1,4,3,1,  1,4,8,0,  4,4,7,8,  0
1
"!"
"HINES, C.O.   (UNIV. OF TORONTO)"
"SECOND ORDER PERTURBATIONS: ENERGY DENSITY AND ENERGY FLUX!"
"NATIONAL CENTER FOR ATMOSPHERIC RESEARCH, BOULDER, COLORADO,"
"TECHNICAL NOTES, NO. 43, SEPTEMBER 1969!"
"FROM THE FULL HYDRODYNAMIC EQUATIONS DEFINING ENERGY DENSITY"
"AND ENERGY FLUX (EV + PV), THE TERM EV REPRESENTS ADVECTED"
"ENERGY AND PV IS THEN CONSIDERED AS BEING PROPAGATED OR"
"TRANSMITTED ENERGY WHICH PASSES FROM ONE FLUID PARCEL TO THE"
"NEXT. FROM THESE EQUATIONS, PERTURBATION EQUATIONS ARE FORMED"
"AND DEDUCED. FIRST AND SECOND ORDER SOLUTIONS ARE PRESENTED."
"EQUATIONS FOR ENERGY DENSITY AND ENERGY FLUX ARE GIVEN."
"SUBJECT HEADING:    ATMOSPHERIC GRAVITY WAVES!"
  1,4,3,0,  3,7,1,8,  0
0
"MGA 21.6-278"
"ALSO IN 'INTERNAL GRAVITY AND ACOUSTIC WAVES', COLLOQUIUM ON"
"INTERNAL GRAVITY AND ACOUSTIC WAVES, BOULDER, JUNE 17 - JULY 26"
"1968.      IN AES LIBRARY.!"
"HINES, C.O.      (UNIV. OF TORONTO)"
"AN EFFECT OF MOLECULAR DISSIPATION IN UPPER ATMOSPHERIC"
"GRAVITY WAVES!"
"JOURNAL OF ATMOSPHERIC AND TERRESTRIAL PHYSICS, 30, 5,"
" 845-849, MAY 1968!"
"!"
  1,4,2,5,    1,9,4,0,  0
3
"MGA 19.12-484!"
"HINES, C.O."
"TIDAL OSCILLATIONS, SHORTER PERIOD GRAVITY WAVES, AND SHEAR"
"WAVES!"
"METEOROLOGICAL MONOGRAPHS 9, 31, 114-121, APRIL 1968!"
"!"
  1,4,0,0,   0
0
"!"
"HINES, C.O."
"APPLICATIONS OF GRAVITY WAVE THEORY TO UPPER ATMOSPHERIC"
"STUDIES!"
"IN: WINDS AND TURBULENCE IN STRATOSPHERE, MESOSPHERE AND IONOSPHERE: PRO-"
"CEEDINGS OF NATO ADVANCED STUDY INSTITUTE, LINDAU, GERMANY,1966. ED. BY"
"KARL RAWER. AMSTERDAM, NORTH-HOLLAND PUB. CO., 1968!"
"!"
  1,4,1,0,  1,9,4,0,  0
0
"!"
"HINES, C.O."
"A POSSIBLE SOURCE OF WAVES IN NOCTILUSCENT CLOUDS!"
"JOURNAL OF THE ATMOSPHERIC SCIENCES 25, 5, 937-942 1968!"
"!"
```

```
1,2,8,0,0
0
"!"
"FULLERTON, C.M."
"AN ANALYTICAL INVESTIGATION OF SINUSOIDAL MICROBAROMETRIC OSCILLATIONS!"
"PH.D. THESIS, NEW MEXICO INSTITUTE OF MINING AND TECHNOLOGY, JUNE 1966!"
"!"
4,10,4,7,0
0
"!"
"FORMAN, M.L."
"FAST FOURIER TRANSFORM TECHNIQUE AND ITS APPLICATIONS TO FOURIER"
"SPECTROSCOPY!"
"J.OPT.SOC.AMER. 56, 7, 978, 1966!"
"!"
6,8,2,3,0
0
"!"
"CURRY, M.J."
"MICROBAROGRAPHIC STUDIES OF THUNDERSTORMS!"
"M.SC. THESIS, UNIVERSITY OF WESTERN ONTARIO, 1968!"
"!"
4,10,2,7,4,4,2,0,4,10,0,1,0
0
"!"
"TOLSTOY, I. AND ENGELHARDT, J."
"NOTE ON LONG GRAVITY WAVES IN LAYERED ATMOSPHERES!"
"J. GEOPHYSICAL RESEARCH 74, 13, 3436-3439, 1969!"
"!"
1,4,5,0,0
0
"MGA 22.1-255!"
"TOLSTOY, I. AND HERRON, T.J."
"ATMOSPHERIC GRAVITY WAVES FROM NUCLEAR EXPLOSIONS!"
"J. ATMOSPHERIC SCIENCES 27, 1, 55-61, 1970!"
"!"
1,4,0,7,0
0
"MGA 21.6-472!"
"TOLSTOY, I. AND PAN, P."
"SIMPLIFIED ATMOSPHERIC MODELS AND THE PROPERTIES OF LONG PERIOD"
"INTERNAL AND SURFACE GRAVITY WAVES!"
"J. ATMOSPHERIC SCIENCES, 27, 1, 31-51, 1970!"
"!"
4,19,7,0,  1,4,4,0;  0
0
"MGA 21.6-271!"
"GOSSARD, E.E."
"GRAVITY WAVES IN THE LOWER TROPOSPHERE OVER SOUTHERN CALIFORNIA!"
"U.S. NAVY ELECTRONICS LABORATORY, SAN DIEGO. RESEARCH AND"
"DEVELOPMENT REPORT; NEL/REPORT NO. 709, 9 AUG. 1956!"
"!"
1,4,2,0, 0
0
"MGA 11.2-179!"
"HAY, D.R. AND POAPS, G.E."
"FRONTAL PERTURBATION OF A TROPOSPHERE SCATTER PATH!"
"CANADIAN J. PHYSICS 37, 11, 1272-1282, 1959!"
"!"
```

```
1,8,11,0, 4,8,0,7, 0
0
"MGA 12.9-103!"
"HITSCHFELD, W.(MCGILL UNIVERSITY)"
"PLUME FORMATION IN THUNDERSTORMS!"
"AMER. GEOPHYS. UNION GEOPHYSICAL MONOGRAPHS, NO. 5, 94-103, 1960!"
"!"
4,4,7,0,  0
0
"MGA 13.1-369!"
```

APPENDIX C

A TYPICAL CODE GROUP SCHEME

This code group system is based on a four-level hierarchy in which each number represents a successively higher degree of specification. This is, of course, only one of the many possible schemes. It is left to the user to adopt this method to his own requirements or to invent an alternate.

FIRST CODE NUMBER		SECOND CODE NUMBER		THIRD CODE NUMBER		FOURTH CODE NUMBER	
NO.	MEANING	NO.	MEANING	NO.	MEANING	NO.	MEANING
1	Waves	1	General	1	Theory	1	Instrumentation
		2	Atmospheric	2	Observation	2	Analysis
		3	Ocean	3	Measurement	3	Laboratory Simulat
		4	Acoustic Gravity	4	Properties	4	Turbulence
		5	Infrasonic	5	Propagation	5	Effects of Medium
		6	Sonic	6	Spectra	6	Computer Programs
		7	Ultrasonic	7	Models	7	Nuclear Explosions
		8	Electromagnetic	8	Generation		
		9	Ionospheric	9	Ray Tracing		
				10	Radiation		
				11	Scattering		
				12	Ducting		
				13	Coupling		
				14	Mode Theory		
2	Acoustics	1	General	1	Theory	1	Instrumentation
		2	Noise	2	Observation	2	Analysis
		3	Vibration	3	Measurement	3	Laboratory Simulat
		4	Architectural	4	Properties	4	Turbulence
		5	Music	5	Propagation	5	Effects of Medium
		6	Psychological Effects	6	Spectra	6	Computer Programs
		7	Physiological Effects	7	Models		
		8	Hearing	8	Generation		
				9	Ray Tracing		
				10	Radiation		
				11	Scattering		
				12	Ducting		
				13	Coupling		
				14	Mode Theory		
				15	Pollution		
				16	SST & Sonic Boom		
				17	Control		
				18	Recording		
				19	Reproduction		

FIRST CODE NUMBER NO.	MEANING	SECOND CODE NUMBER NO.	MEANING	THIRD CODE NUMBER NO.	MEANING	FOURTH CODE NUMBER NO.	MEANING
3	Fluid Dynamics	1	General	1	Theory	1	Instrumentation
		2	Drops	2	Observation	2	Analysis
		3	Wakes	3	Measurement	3	Laboratory Simulati
		4	Flow	4	Properties	4	Turbulent
		5	Drag--sphere			5	Effects of Medium
		6	Drag--non-sphere	6	Spectra	6	Computer programs
		7	Basic Equations	7	Generation	7	Laminar
						8	Perturbation Metho
4	Meteorology (Atmospheric Science)	1	General	1	Theory	1	Instrumentation
		2	Fair Weather	2	Observation	2	Analysis
		3	Severe Storms	3	Measurement	3	Laboratory Simulati
		4	Thunderstorms	4	Properties		
		5	Rain	5	Propagation or Movement	5	Effects of Medium
		6	Hail	6	Spectra	6	Computer programs
		7	Fog	7	Models	7	Perturbations
		8	Fronts	8	Generation or Formation	8	Radiosonde probe
		9	Jet Stream	9	Distribution		
		10	Pressure	10	Forecasting		
		11	Wind	11	Modification		
		12	Temperature	12	Prevention		
		13	Humidity				
		14	Lightning				
		15	Turbulence				
		16	Stability				
		17	Convection				
		18	Cloud				
		19	Profiles				

FIRST CODE NUMBER		SECOND CODE NUMBER		THIRD CODE NUMBER		FOURTH CODE NUMBER	
NO.	MEANING	NO.	MEANING	NO.	MEANING	NO.	MEANING
5	Electronics	1	General	1	Theory	1	DC
		2	Amplifiers, low signal	2	Design	2	Audio
		3	Amplifiers, large signal	3	Circuit Analysis	3	RF
		4	Power Supplies	4	Circuit Diagrams	4	Instrumentation
		5	Power Converters	5	Construction		
		6	A/D and D/A conversion	6	Testing		
		7	Oscillators				
		8	Analogue Circuits				
		9	Digital Circuits				
		10	Control Circuits				
		11	Test Equipment				
		12	Recorders				
		13	Loudspeakers and Enclosures				
		14	Actuators				
		15	Interfaces				

FIRST CODE NUMBER		SECOND CODE NUMBER		THIRD CODE NUMBER		FOURTH CODE NUMBER	
NO.	MEANING	NO.	MEANING	NO.	MEANING	NO.	MEANING
6	Mathematics and Data Processing	1	General	1	Theory	1	Analytical
		2	Vector Methods	2	Technique	2	Numerical
		3	Tensor Methods	3	Application	3	Computer-based
		4	Complex Variables	4	Solution	4	Algorithm
		5	Differential Equations	5	Introduction to		
		6	Integral Equation				
		7	Differential Geometry				
		8	Fourier Transforms				
		9	Laplace Transforms				
		10	Topology				
		11	Numerical Methods				
		12	Statistics				

APPENDIX D

DE-BUGGING A DATA FILE

If, during execution, the program INFORM encounters faulty data, an error message is generated indicating in which part of the program the problem occurred. For example, the message BAD DATA IN LINE 1140 would indicate that the computer was unable to successfully read an Author field. However, the computer does not indicate where in the data file the fault lies.

The program given in this Appendix was written to assist the user in locating typographical errors in a data file. This program named CHECK, contains input statements identical to those used by INFORM. Its sole function, is, however, to read the data file, line by line, incrementing a counter as each line is successfully read and writing the counter value into a disk file. If the program encounters faulty data and terminates, the user need only examine the disk file (named CHECK.OPT) to determine where the fault lies. If, for example, the number in this file is 493, the user should look for faulty data in the few lines preceding the 493rd line in the data file. Possible faults include a missing quotation mark at either end of a line of text, a missing exclamation mark at the end of a field, or quotation marks within a line of text.

If no bad data are found, CHECK indicates that the file was successfully read. The file may then be assumed to be readable by INFORM.

The program CHECK follows as figure D-1. Figure D-2 indicates a portion of a data file named TEST.DAT in which quotation marks appear within the text in the 11th line. Figure D-3 shows that CHECK correctly locates the faulty data as occurring just before line 12.

```
10      MARGIN 80
20      MARGIN ALL 80
30      PRINT "  WHAT IS THE NAME OF YOUR DATA FILE"
40      PRINT "  (ENCLOSE IN DOUBLE QUOTES)    ";
50      INPUT F1$
60      FILE #1, F1$
70      FILE #2, "CHECK.OP1"
75      SCRATCH #2
76      PRINT
77      PRINT
78      PRINT "   READING ...   "
79      PRINT
80      LET N=0
90      FOR I=1 TO 5000
100         IF END #1 THEN 510
110         LET N=N+1
115         SCRATCH #2
120         PRINT USING #2, 515, N
130     INPUT #1, Z$
140         FOR J1=1 TO 5
150           LET N=N+1
155           SCRATCH #2
160           PRINT USING #2, 515, N
170           INPUT #1, Z$
180           IF RIGHT$(Z$,1)="!" THEN 200
190           NEXT J1
200         FOR J1=1 TO 5
210           LET N=N+1
215           SCRATCH #2
220           PRINT USING #2, 515, N
230           INPUT #1, Z$
240           IF RIGHT$(Z$,1)="!" THEN 260
250           NEXT J1
260         FOR J1=1 TO 20
270           LET N=N+1
275           SCRATCH #2
280           PRINT USING #2, 515, N
290           INPUT #1, Z$
300           IF RIGHT$(Z$,1)="!" THEN 320
310           NEXT J1
320         LET N=N+1
325           SCRATCH #2
330           PRINT USING #2, 515, N
340         FOR J2=1 TO 6
350           INPUT #1, A
360           IF A=0 THEN 410
370           FOR J3=2 TO 4
380             INPUT #1, A
390             NEXT J3
400           NEXT J2
410         LET N=N+1
415           SCRATCH #2
420           PRINT USING #2, 515, N
430         INPUT #1, A
440         FOR J1=1 TO 5
450           LET N=N+1
455           SCRATCH #2
460           PRINT USING #2, 515, N
```

```
470           INPUT #1, Z$
480           IF RIGHT$(Z$,1)="!" THEN 500
490           NEXT J1
500         NEXT I
510     PRINT "  FILE SUCCESSFULLY READ:  ";N;" LINES"
515     :  LAST LINE READ WAS #####
520     END
```

1130

```
TYPE TEST.DAT
"ATKINSON, B.W."
"STRUCTURE OF THE THUNDER ATMOSPHERE, SOUTH-EAST ENGLAND, 1951-60!"
"WEATHER 22, 8, 335-345, 1967!"
"MEAN TEMPERATURE AND HUMIDITY PROFILES WERE CALCULATED USING RADIO-SONDE"
"DATA FROM CRAWLEY, OR LARKHILL, IN SOUTHERN ENGLAND, FOR EACH OCCASION"
"WHEN THUNDERSTORMS WERE OBSERVED WITHIN 100 MI OF THE RADIOSONDE STATION."
"THESE MEAN 'THUNDER ATMOSPHERE' PROFILES ARE PRESENTED AND COMPARED WITH"
"THE OVERALL MEAN PROFILES. THE DATA WERE DIVIDED ACCORDING TO SYNOPTIC"
"TYPE AND MEAN VALUES OF VARIOUS TEMPERATURE AND HUMIDITY PARAMETERS WERE"
"CALCULATED. SOME OF THESE VALUES ARE TABULATED HERE. STABILITY PATTERNS"
"ARE ALSO CONSIDERED AND THE "MEAN" PATTERN OF STABILITY FOR THUNDERSTORM"
"OCCASIONS IN THE DIFFERENT SEASONS AND FOR DIFFERENT WEATHER TYPES ARE"
"BRIEFLY DISCUSSED.   SUBJECT HEADINGS: 1. THUNDERSTORM ANALYSIS  .....!"
 4,4,4,8,  4,12,9,8,  4,13,9,8,  4,16,3,0,  0
9
"MGA 19.1-191!"
"REDDY, C.A.  -(NCAR, BOULDER)"
^C
```

Figure D-2

A portion of a data file containing faulty data

1130

```
R BASIC

NEW OR OLD--OLD
OLD FILE NAME--CHECK

READY
RUN

CHECK          13:21          14-OCT-73


   WHAT IS THE NAME OF YOUR DATA FILE
   (ENCLOSE IN DOUBLE QUOTES)     ?"TEST.DAT"

    READING ...

? BAD DATA IN LINE 290

RUN TIME:  0.86 SECS.

READY
SYS

EXIT
^C

.TYPE CHECK.OPT
  LAST LINE READ WAS     12

.
```

Figure D-3
The Operation of CHECK

TEC-798 UDC: 016:681.14
14 December 1973

CANADA
Environment - Atmospheric Environment Service
4905 Dufferin Street, Downsview, Ontario

A BASIC Program for the Manipulation of
Bibliographic Information by a
Small Research Group
by M.J. Curry

41 pps.

Subjec Reference: 1. Computerized Search
2. Scientific Literature

TEC-798 UDC: 016:681.14
14 December 1973

CANADA
Environment - Atmospheric Environment Service
4905 Dufferin Street, Downsview, Ontario

A BASIC Program for the Manipulation of
Bibliographic Information by a
Small Research Group
by M.J. Curry

41 pps.

Subject Reference: 1. Computerized Search
2. Scientific Literature

TEC-798 UDC; 016:681.14
14 December 1973

CANADA
Environment - Atmospheric Environment Service
4905 Dufferin Street, Downsview, Ontario

A BASIC Program for the Manipulation of
Bibliographic Information by a
Small Research Group
by M.J. Curry

41 pps.

Subject Reference: 1. Computerized Search
2. Scientific Literature

TEC-798 UDC: 016: 681.14
14 December 1973

CANADA
Environment - Atmospheric Environment Service
4905 Dufferin Street, Downsview, Ontario

A BASIC Program for the Manipulation of
Bibliographic Information by a
Small Research Group
by M.J. Curry

41 pps.

Subject Reference: 1. Computerized Search
2. Scientific Literature

ABSTRACT: A computer program is described, written in the BASIC language, which performs search-and-retrieve operations on a personalized bibliographic data file using a time-sharing computer or a mini-computer. It is shown that the program provides a convenient and economical way of cataloguing and accessing reference literature of interest to an individual or a small research group. The design philosophy of the program is briefly discussed, and examples of its operation are given.

ABSTRACT: A computer program is described, written in the BASIC language, which performs search-and-retrieve operations on a personalized bibliographic data file using a time-sharing computer or a mini-computer. It is shown that the program provides a convenient and economical way of cataloguing and accessing reference literature of interest to an individual or a small research group. The design philosophy of the program is briefly discussed, and examples of its operation are given.

ABSTRACT: A computer program is described, written in the BASIC language, which performs search-and-retrieve operations on a personalized bibliographic data file using a time-sharing computer or a mini-computer. It is shown that the program provides a convenient and economical way of cataloguing and accessing reference literature of interest to an individual or a small research group. The design philosophy of the program is briefly discussed, and examples of its operation are given.

ABSTRACT: A computer program is described, written in the BASIC language, which performs search-and-retrieve operations on a personalized bibliographic data file using a time-sharing computer or a mini-computer. It is shown that the program provides a convenient and economical way of cataloguing and accessing reference literature of interest to an individual or a small research group. The design philosophy of the program is briefly discussed, and examples of its operation are given.